

# Diagonalization

Worksheet by Russell Blyth

```
> restart:with(plots):with(LinearAlgebra):  
  randomize(52):  
Warning, the name changecoords has been redefined
```

## Outline

Goals of this worksheet:

- 1) Construct a diagonalizable matrix which has specified eigenvalues and eigenvectors.
- 2) Compute powers of diagonalizable matrices with ease!

```
>
```

```
>
```

## Diagonalizing matrices

We begin by applying the diagonalization process to a random 2x2 matrix.

```
> A := RandomMatrix(2,2,generator=rand(-10..10));
```

$$A := \begin{bmatrix} 0 & 1 \\ -7 & -7 \end{bmatrix} \quad (2.1)$$

```
> Eigenvectors(A);
```

$$\begin{bmatrix} -\frac{7}{2} + \frac{1}{2}\sqrt{21} \\ -\frac{7}{2} - \frac{1}{2}\sqrt{21} \end{bmatrix}, \begin{bmatrix} \frac{1}{-\frac{7}{2} + \frac{1}{2}\sqrt{21}} & \frac{1}{-\frac{7}{2} - \frac{1}{2}\sqrt{21}} \\ 1 & 1 \end{bmatrix} \quad (2.2)$$

The eigenvectors appear in the first vector and corresponding eigenvectors appear as columns in the second matrix. If you prefer to work with only real values and the matrix above has complex-valued eigenvalues and eigenvectors, re-execute the two lines above until the eigenvalues (and eigenvectors) are real. Likewise if there is only one nonzero eigenvector displayed, or if the matrix is already diagonal.

Extract the matrix of eigenvectors and find its inverse.

```
> Q := Eigenvectors(A)[2];  
Q2 := MatrixInverse(Q);
```

$$Q := \begin{bmatrix} \frac{1}{-\frac{7}{2} + \frac{1}{2}\sqrt{21}} & \frac{1}{-\frac{7}{2} - \frac{1}{2}\sqrt{21}} \\ 1 & 1 \end{bmatrix}$$

(2.3)

$$Q2 := \begin{bmatrix} \frac{1}{84} \sqrt{21} (-7 + \sqrt{21}) (7 + \sqrt{21}) & \frac{1}{42} (-7 + \sqrt{21}) \sqrt{21} \\ -\frac{1}{84} \sqrt{21} (-7 + \sqrt{21}) (7 + \sqrt{21}) & \frac{1}{42} (7 + \sqrt{21}) \sqrt{21} \end{bmatrix} \quad (2.3)$$

Diagonalize the matrix A using the matrix of eigenvectors. We also confirm that the diagonal entries of the diagonalized matrix are just the eigenvalues of A by computing numerical values.

```
> expand(simplify(Q2.A.Q));
evalf(simplify(Q2.A.Q));
evalf(Eigenvectors(A)[1]);
```

$$\begin{bmatrix} -\frac{1}{6} (-3 + \sqrt{21}) \sqrt{21} & 0 \\ 0 & -\frac{1}{6} (3 + \sqrt{21}) \sqrt{21} \end{bmatrix}$$

$$\begin{bmatrix} -1.208712153 & 0. \\ 0. & -5.791287849 \end{bmatrix}$$

$$\begin{bmatrix} -1.208712152 \\ -5.791287848 \end{bmatrix} \quad (2.4)$$

```
>
```

We attempt the same computation with a random 3x3 matrix.

```
> A := RandomMatrix(3,3,generator=rand(-10..10));
MatrixInverse(A);
evalf(Eigenvectors(A)[1]);
Q := evalf(Eigenvectors(A)[2]);
Q2 := MatrixInverse(Q);
Q2.A.Q;
```

$$A := \begin{bmatrix} -9 & -7 & -9 \\ -5 & 7 & 6 \\ 1 & 6 & 4 \end{bmatrix}$$

$$\begin{bmatrix} \frac{-8}{223} & \frac{-26}{223} & \frac{21}{223} \\ \frac{26}{223} & \frac{-27}{223} & \frac{99}{223} \\ \frac{-37}{223} & \frac{47}{223} & \frac{-98}{223} \end{bmatrix}$$

$$\begin{bmatrix} 13.27975870 - 1.10^{-9} I \\ -9.514897455 - 7.32050808 \cdot 10^{-10} I \\ -1.764861245 + 2.732050808 \cdot 10^{-9} I \end{bmatrix}$$

$$Q := \begin{bmatrix} [-0.9390564544 - 1.469066096 \cdot 10^{-9} I, -4.024455727 + 6.104690949 \cdot 10^{-9} I, \\ -0.3747777876 + 6.492744342 \cdot 10^{-10} I], [1.703135890 + 4.266389349 \cdot 10^{-9} I, \\ -1.581740287 - 1.026136811 \cdot 10^{-11} I, -0.8983472445 - 3.794229165 \cdot 10^{-10} I], [1., 1., 1.] \\ ] \end{bmatrix}$$

$$Q2 := \begin{bmatrix} [-0.0691679380805629157 + 6.53866792667350341 \cdot 10^{-11} I, \\ 0.369393134005187972 - 7.01813541269412182 \cdot 10^{-10} I, \\ 0.305920697264092711 - 4.20901591856703783 \cdot 10^{-10} I], [ \\ -0.263302686996150780 - 3.63539169931479452 \cdot 10^{-10} I, \\ -0.0571120708847500242 - 1.91264046916137105 \cdot 10^{-10} I, \\ -0.149986470008556528 - 1.58781860711571970 \cdot 10^{-10} I], [ \\ 0.332470625076713600 + 2.98152490664744391 \cdot 10^{-10} I, \\ -0.312281063120437962 + 8.93077588185549340 \cdot 10^{-10} I, \\ 0.844065772744463704 + 5.79683452568275728 \cdot 10^{-10} I] \\ ] \end{bmatrix}$$

$$\begin{bmatrix} [13.2797587025850632 + 4.34085391870130262 \cdot 10^{-17} I, \\ 7.65352226039794914 \cdot 10^{-11} - 9.71142083658384950 \cdot 10^{-9} I, \\ -7.07253189347056832 \cdot 10^{-9} - 2.78423550172641621 \cdot 10^{-9} I], [ \\ -1.38990365883273625 \cdot 10^{-8} - 3.26298026826240414 \cdot 10^{-9} I, \\ -9.51489745491762839 + 7.06710173738668290 \cdot 10^{-18} I, \\ 1.39964928536073784 \cdot 10^{-10} + 1.15697248740205701 \cdot 10^{-9} I], [ \\ 1.96913972905221613 \cdot 10^{-7} + 2.73922502228538638 \cdot 10^{-8} I, \end{bmatrix} \quad (2.5)$$

$$\begin{bmatrix} 5.84109050016934362 \cdot 10^{-9} + 1.57545435698567522 \cdot 10^{-8} I, \\ -1.76486124766743391 - 5.04756395166588515 \cdot 10^{-17} I \end{bmatrix}$$

>

**Important: Read through this section before executing the next block**

Note that the Maple procedure for finding the roots of cubics does not always recognize when the eigenvalues and/or eigenvectors turn out to be real - rounding error often produces complex values with very small imaginary parts. We can help the numerical evaluation by dropping the imaginary parts if all the entries should be real but appear with very small (nonzero) imaginary parts. Execute the following block of code only if the three eigenvalues are presumed to be real but very small imaginary parts appear as eigenvalues and eigenvalue entries.

```
> Evals := Eigenvectors(A)[1]:
Evecs := Eigenvectors(A)[2]:
Evals := map(Re,evalf(Evals));
Evecs := map(Re,evalf(Evecs));
Evecs2 := MatrixInverse(Evecs);
evalf(simplify(Evecs2.A.Evecs));
```

$$Evals := \begin{bmatrix} 13.27975870 \\ -9.514897455 \\ -1.764861245 \end{bmatrix}$$

$$Evecs := \begin{bmatrix} -.9390564544 & -4.024455727 & -.3747777876 \\ 1.703135890 & -1.581740287 & -.8983472445 \\ 1. & 1. & 1. \end{bmatrix}$$

$$Evecs2 := \begin{bmatrix} [-0.0691679380805628880, 0.369393134005187916, 0.305920697264092766], [ \\ -0.263302686996150780, -0.0571120708847500242, -0.149986470008556528], [ \\ 0.332470625076713600, -0.312281063120437962, 0.844065772744463704] \end{bmatrix}$$

$$\begin{bmatrix} 13.27975870 & 7.653522260 \cdot 10^{-11} & -7.072531893 \cdot 10^{-9} \\ -1.389903659 \cdot 10^{-8} & -9.514897455 & 1.399649285 \cdot 10^{-10} \\ 1.969139731 \cdot 10^{-7} & 5.841090500 \cdot 10^{-9} & -1.764861248 \end{bmatrix}$$

(2.6)

>

Note the eigenvalues appear on the diagonal of Q2.A.Q (and any off-diagonal entries are within

rounding error of zero).

## Constructing easily diagonalizable matrices

How might a professor find a diagonalizable matrix  $A$  to use as an example or on a test question?

We see here how to use the equation  $A = QDQ^{-1}$  for this purpose, where  $D$  is diagonal with entries that are eigenvalues and  $Q$  is the matrix whose columns are eigenvectors of  $A$ . We first randomly choose the eigenvalues, and then the eigenvectors corresponding to each of these eigenvalues. (Note we cannot use the variable  $D$ , which is a Maple system variable)

```
> DD := RandomMatrix(3,3,generator=rand(-5..5), outputoptions=[shape=diagonal]);
```

$$DD := \begin{bmatrix} -1 & 0 & 0 \\ 0 & 5 & 0 \\ 0 & 0 & -5 \end{bmatrix} \quad (3.1)$$

```
> Q1 := RandomVector(3,generator=rand(-3..3));  
Q2 := RandomVector(3,generator=rand(-3..3));  
Q3 := RandomVector(3,generator=rand(-3..3));
```

$$Q1 := \begin{bmatrix} 0 \\ -2 \\ -1 \end{bmatrix}$$
$$Q2 := \begin{bmatrix} 2 \\ -1 \\ 3 \end{bmatrix}$$
$$Q3 := \begin{bmatrix} -2 \\ 3 \\ 2 \end{bmatrix} \quad (3.2)$$

Construct the matrix  $Q$  by assembling  $Q1$ ,  $Q2$ ,  $Q3$  as its columns

```
> Q := <Q1|Q2|Q3>;
```

$$Q := \begin{bmatrix} 0 & 2 & -2 \\ -2 & -1 & 3 \\ -1 & 3 & 2 \end{bmatrix} \quad (3.3)$$

Check that  $\{Q1, Q2, Q3\}$  is independent.

```
> Rank(Q);
```

$$3 \quad (3.4)$$

If the rank of  $Q$  is not three, then the set  $\{Q1, Q2, Q3\}$  is not independent - go back and compute a new set.

Now compute the matrix  $A$

```
> A := Q . DD . Q^(-1);
```

$$A := \begin{bmatrix} \frac{-15}{4} & \frac{-5}{2} & 5 \\ \frac{39}{8} & \frac{5}{4} & \frac{-9}{2} \\ \frac{37}{8} & \frac{-5}{4} & \frac{3}{2} \end{bmatrix} \quad (3.5)$$

Check that the matrix A has the expected eigenvalues and eigenvectors

> **Eigenvectors(A);**

$$\begin{bmatrix} -5 \\ -1 \\ 5 \end{bmatrix}, \begin{bmatrix} -1 & 0 & \frac{2}{3} \\ \frac{3}{2} & 2 & \frac{-1}{3} \\ 1 & 1 & 1 \end{bmatrix} \quad (3.6)$$

Let's also check the characteristic polynomial of A.

> **chp := CharacteristicPolynomial(A, x);**

$$chp := x^3 + x^2 - 25x - 25 \quad (3.7)$$

> **factor(chp);**

$$(x - 5)(x + 5)(x + 1) \quad (3.8)$$

### Exercises:

1) Find a 3 x 3 matrix B which has the eigenvectors and eigenvalues of your choice. Make all your eigenvalues nonzero, and all entries of your eigenvectors nonzero. (Don't reuse the variables DD or Q)

>

2) The matrices A and B probably have fractional entries. How could the professor make sure that a matrix computed this way has only integer entries?

>

>

## Computing powers of matrices using diagonalization

Compute a high power of A using the diagonal matrix DD. First we choose an appropriate power:

> **power := rand(8..20)();**

$$power := 18 \quad (4.1)$$

Raise the matrix DD to this power.

> **powerDD := DD^power;**

$$powerDD := \begin{bmatrix} 1 & & 0 & & 0 \\ 0 & 3814697265625 & & & 0 \\ & & & & \\ 0 & & 0 & 3814697265625 & \end{bmatrix} \quad (4.2)$$

A to the given power is now computed using Q

```
> powerA := Q . powerDD . Q^(-1);
```

$$powerA := \begin{bmatrix} 3814697265625 & 0 & 0 \\ -5245208740233 & -953674316405 & 1907348632812 \\ \frac{-5245208740233}{2} & -2384185791015 & 4768371582031 \end{bmatrix} \quad (4.3)$$

Check by direct calculation:

```
> A^power;
```

$$\begin{bmatrix} 3814697265625 & 0 & 0 \\ -5245208740233 & -953674316405 & 1907348632812 \\ \frac{-5245208740233}{2} & -2384185791015 & 4768371582031 \end{bmatrix} \quad (4.4)$$

**Exercise:**

3) Compute some power of B between 8 and 20 using the same technique.

```
>
```

```
>
```